THE SIMPLICIAL COMPLEXES PACKAGE FOR MACAULAY2

BEN HERSEY, GREGORY G. SMITH, AND ALEXANDRE ZOTINE

ABSTRACT. This article demonstrates some of the updated features of the `SimplicialComplexes` package in *Macaulay2*. todo

2. COMBINATORIAL TOPOLOGY

Lutz has provided a database enumerating all of the 2 and 3-manifolds having 10 or less vertices. We have implemented these databases into the package—however we have excluded the database of 3-manifolds with 10 vertices, due to the large number of examples causing long loading times.

Example 2.1. These databases can be used to find nice testbeds of examples: for instance, we can search for simplicial maps

```
i2 : R = ZZ[a..i];
i3 : S = ZZ[x_0..x_6];
i4 : Γ = smallManifold(2,7,1,S);
i5 : maplist = flatten for i to 2 list (
     for j in subsets(toList(R_0..R_8),7) list (
         phi := map(smallManifold(3,9,i,R),Γ,j);
         if isWellDefined phi then phi else continue
         )
     );
i6 : maplist_0
o6 = | a b e f g h i |
```

By construction, all of these maps should be inclusions.

```
i7 : isInjective\maplist
o7 = {true, true, true, true, true, true}
o7 : List
```

◇

The database also contains many triangulations of various interesting surfaces, such as the torus, Klein bottle, and real projective plane. Here are the smallest indices (and hence minimal triangulations of) these surfaces in the database

Example 2.2.

```
i8 : Torus = smallManifold(2, 7, 6, R);
i9 : KleinBottle = smallManifold(2, 8, 12, R);
i10 : RP2 = smallManifold(2, 6, 1, R);
```

Date: May 28, 2021.
todo grant stuff *Mathematics Subject Classification*. todo.

We can check that these are the right surfaces by computing their homology. Theorems 6.2, 6.3, and 6.4 from Munkres confirm that they match [Mun18].

```
i11 : for i to 2 list prune HH_i Torus
            2    1
o11 = {0, ZZ , ZZ }
o11 : List
i12 : for i to 2 list prune HH_i KleinBottle
o12 = {0, cokernel | 2 |, 0}
                   | 0 |
o12 : List
i13 : for i to 2 list prune HH_i RP2
o13 = {0, cokernel | 2 |, 0}
o13 : List
```

We can explicitly identify the generators of the homology for the torus.

```
i0 : TODO. Ben's code
```

◇

3. Stanley-Reisner Theory

Let Δ be an abstract simplicial complex with vertex set $V = \{v_0, v_1, \dots, v_{n-1}\}$, let k be a commutative ring, and let $S = k[x_0, x_1, \dots, x_n]$. The **Stanley-Reisner ideal**, or **facet ideal** of Δ is defined to be square-free monomial ideal

$$I_\Delta := \left(\prod_{j=1}^{k} x_{i_j} \;\middle|\; \{v_{i_1}, v_{i_2}, \dots, v_{i_k}\} \not\subset \Delta \right) \subset S,$$

and the **Stanley-Reisner ring** corresponding to Δ is $k[\Delta] = S/I_\Delta$. This correspondence between simplicial complexes and square-free monomial ideals is one-to-one. Stanley-Reisner theory connects homological properties of $k[\Delta]$ to combinatorial and topological properties of Δ. A survey of results can be found in [BH93, Sta96, MS05].

If $I = (m_1, \dots, m_q) \subset S$ is a monomial ideal, with minimal generators $m_i = \prod x_j^{a_{i_j}}$, then the **Alexander dual** of I is defined to be $I^* := \bigcap_{i=1}^{q} (x_0^{a_{i_1}},\, x_1^{a_{i_2}}, \dots,\, x_{n-1}^{a_{i_{n-1}}})$. If $I = I_\Delta$ for some simplicial complex Δ, then I^* is also a square-free monomial ideal and is the Stanley-Reisner ideal of a simplicial complex Δ^*, which we call the **Alexander dual** complex to Δ. There is also a combinatorial description of Δ^*, given by $\Delta^* = \{F \subset V \mid V \setminus F \not\in \Delta\}$. One of the attractive features of Alexander duality is the relationship between the cohomology of Δ and the homology of Δ^*. More specifically, if Δ is a simplicial complex on n vertices, then $\widetilde{H}_{i-1}(\Delta^*) = \widetilde{H}^{n-2-i}(\Delta)$ for all $i \in \mathbb{Z}$, see [MS05, Theorem 5.6].

Example 3.1. Consider the simplicial complex $\bowtie$, depicted in Figure 1 The Stanley-Reisner ideal of $\bowtie$ is $I_{\bowtie} = (x_0x_1x_2,\; x_0x_3,\; x_1x_3,\; x_0x_4,\; x_1x_4,\; x_2x_3x_4)$. We can exhibit the correspondence between Δ and I_Δ using the methods `simplicialComplex` and `ideal`.



FIGURE 1. The simplicial complex ⋈ (left) and its Alexander dual ⋈* (right).

```
i28 : S = QQ[x_0..x_4];
i29 : I⋈ = monomialIdeal(x_0*x_1*x_2,x_0*x_3,x_1*x_3,x_0*x_4,x_1*x_4,x_2*x_3*x_4);
o29 : MonomialIdeal of S
i30 : ⋈ = simplicialComplex I⋈
o30 = simplicialComplex | x_3x_4 x_2x_4 x_2x_3 x_1x_2 x_0x_2 x_0x_1 |
o30 : SimplicialComplex
i31 : I⋈ == ideal ⋈
o31 = true
```

We can use the dual method to compute the Alexander dual of Δ.

```
i133 : dual ⋈
o133 = simplicialComplex | x_1x_2x_4 x_0x_2x_4 x_1x_2x_3 x_0x_2x_3 |
o133 : SimplicialComplex
```

which is the simplicial complex By the definition of the Alexander dual, we know that $(I_\Delta)^* = I_{\Delta^*}$. We can verify this directly.

```
i134 : dual(monomialIdeal Δ) == monomialIdeal dual Δ
o134 = true
```

We can also verify the combinatorial description of Δ^* by showing that the minimal generators of I_Δ correspond the complements of the facets of Δ^*,

```
i140 : dualFacets = first entries facets dual Δ
o140 = {x x x , x x x , x x x , x x x }
         1 2 4   0 2 4   1 2 3   0 2 3
o140 : List
i141 : sort first entries gens IΔ == sort for F in dualFacets list(
           product for v in vertices Δ list(
               if member(v, support F) then continue else v)
               )
           )
o141 = true
```

Finally, we exhibit the isomorphisms between the cohomology of Δ and the homology of Δ^*.

```
i94 : all(-1..5, i -> all(-1..5, i -> prune HH^(3-i) Δ == prune HH_(i-1) dual Δ)
o94 = true
```

◇

For a face $F \in \Delta$, we define the **link** of F, is the subcomplex of Δ defined by

$$\operatorname{link}_\Delta(F) := \{G \in \Delta \mid F \cup G \in \Delta \text{ and } F \cap G = \varnothing\}.$$

We can now exhibit a more substantive result of Stanley-Reisner theory, which is the "dual version" of Hochster's formula, see [MS05, Corollary 1.40]. This formula allows us to compute the multigraded Betti numbers of I_Δ, which are defined to be $\beta_{i,m}(I_\Delta) = \big(\operatorname{Tor}_i^S(I_\Delta, k)\big)_m$. For a subset $F \subset V$, we will use the notation $\beta_{i,F}$ to refer to the betti number in homological degree i and multidegree $(a_1, \dots, a_n) \in \mathbb{Z}^n$, where $a_i = 1$ if $v_i \in F$ and $a_i = 0$ otherwise.

Theorem 3.2 (Hochster's Formula, dual version)**.** *Let Δ be a simplicial complex with vertex set $V = \{v_0, v_1, \dots, v_{n-1}\}$. The nonzero multigraded Betti numbers of I_Δ and S/I_Δ lie in squarefree degrees. Moreover, if $F \subset V$, then*

$$\beta_{i,F}(I_\Delta) = \beta_{i+1,F}(S/I_\Delta) = \dim_k \Big(\widetilde{H}_{i-1}\big(\operatorname{link}_{\Delta^*}(V \setminus F)\,;\, k\big)\Big).$$

Example 3.3. In Example **??**, we computed the Alexander dual of the figure-8 complex. We can use the `link` method to compute the links of various faces. For example, we compute the link of the central vertex v_2, whose link is a square.

```
i27 : link(dual Δ, x_2)
o27 = simplicialComplex | x_1x_4 x_0x_4 x_1x_3 x_0x_3 |
o27 : SimplicialComplex
```

We can also construct a function that computes the multigraded betti numbers $\beta_{i,F}(S/I_\Delta)$.

```
hochster = (i, m) -> (
    G := product select(vertices Δ, v -> not member(v, support m));
    rank (homology link(dual Δ, G_S))_i
    )
```

Since we have a bound on the nonzero betti numbers of I_Δ, we can collect them into a matrix

```
i94 : V = vertices Δ;
i95 : squarefreeMonomials = unique sort apply(remove(subsets V, 0), m -> lcm m);
i96 : matrix for i to length res IΔ - 1 list (
          for F in squarefreeMonomials list hochster(i-1,F)
          )
o96 = | 0 0 0 0 0 0 0 1 1 0 1 1 0 0 0 1 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 |
      | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 1 0 0 1 0 0 1 0 1 1 0 1 1 0 |
      | 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 2 |
              3        31
o96 : Matrix ZZ  <--- ZZ
```

where the row are indexed by the homological degree (starting at 0), and the columns are indexed by the squarefree multidegrees. ◇

We say that a simplicial complex Δ is **pure** if all of the facets of Δ have the same dimension. We say that **shellable** if we can order the facets $F_1, \dots, F_m$ of Δ so that $\langle F_i \rangle \cap \langle F_1, F_2, \dots, F_{i-1} \rangle$ is a pure, codimension 1, simplicial complex. We say that a Δ is **Cohen-Macaulay** if $k[\Delta]$ is a Cohen-Macaulay ring. It is known that every shellable simplicial complex is Cohen-Macaulay

[BH93, Theorem 5.1.13] and every Cohen-Macaulay simplicial complex is pure [BH93, Corollary 5.1.5].

For Stanley-Reisner rings, we can use the combinatorics of Δ to determine if $k[\Delta]$ is a Cohen-Macaulay ring. Specifically, $k[\Delta]$ is Cohen-Macaulay if an only if $\widetilde{H}_i(\mathrm{link}_\Delta(F);k) = 0$ for all $F \in \Delta$ and $i <$ dim $\mathrm{Link}_\Delta(F)$, see [BH93, Corollary 5.3.9]. We can also relate the Cohen-Macaulay property to the h-vector of Δ. We can write the Hilbert series of $k[\Delta]$ as a rational function

$$\sum \dim\, k[\Delta]_i\, t^i = \frac{h_0 + h_1 t + \cdots h_k t^k}{(1-t)^d}$$

where $d = \dim \Delta + 1$, and $0 \le k \le d$ is such that $h_k \neq 0$. If Δ is a d-dimensional Cohen-Macaulay complex with n vertices, then [BH93, Lemma 5.1.10] proves that

$$0 \le h_i \le \binom{n-d+i}{i}$$

for $i = 0, 1, \ldots, d+1$. We can also compute the h-vector using the combinatorics of Δ. Specifically, if Δ is a d-dimensional simplicial complex, then [BH93, Lemma 5.18] tells us that

$$h_j = \sum_{i=0}^{j} (-1)^{j-1} \binom{d+1-i}{j-1} f_{i-1}$$

for $j = 0, 1, \ldots, d+1$.

Example 3.4. The figure 8 complex introduced in example 3.1 is Shellable and Cohen-Macaulay. We can verify the Cohen-Macaulay property using [BH93, Corollary 5.3.9].

```
i43 : faceList = flatten for i from -1 to dim Δ list (faces Δ)#i
o43 = {1, x , x , x , x , x , x x , x x , x x , x x , x x , x x }
           0   1   2   3   4   0 1   0 2   1 2   2 3   2 4   3 4
o43 : List
i44 : all(faceList , F -> all(0..dim(link(⋈,F))-1, i -> HH_i link(Δ,F) == 0))
o44 = true
```

This simplicial complex is also shellable, but this package does not provide a method for determining this quality. However, such a method is available in the `SimplicialDecomposability` package in `Macaulay2`, see [Coo10].

We instead consider the "bowtie" simplicial complex, which we can obtain by filling in the facets of the 2-faces of the figure 8 simplicial complex, see Figure ◇

4. Resolutions Of Monomial Ideals

Acknowledgements. We thank todo .

partially supported by the Natural Sciences and Engineering Research Council of Canada (NSERC).



References

[BH93] Winfried Bruns and Jürgen Herzog, *Cohen-Macaulay rings*, Cambridge Studies in Advanced Mathematics, vol. 39, Cambridge University Press, Cambridge, 1993.

[Sta96] R.P. Stanley, *Combinatorics and Commutative Algebra*, 2nd ed., Progress in Mathematics, vol. 41, Birkhäuser Boston, 1996.

[MS05] Ezra Miller and Bernd Sturmfels, *Combinatorial Commutative Algebra*, Graduate Texts in Mathematics, vol. 227, Springer-Verlag New York, 2005.

[Pee11] Irena Peeva, *Graded Syzygies*, Algebra and Applications, vol. 14, Springer-Verlag London, 2011.

[Mun18] James R. Munkres, *Elements Of Algebraic Topology*, CRC Press, 2018.

[M2] Daniel R. Grayson and Michael E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at `http://www.math.uiuc.edu/Macaulay2/`.

[ÀMFRG20] Josep Àlvarez Montaner, Oscar Fernández-Ramos, and Philippe Gimenez, *Pruned cellular free resolutions of monomial ideals*, J. Algebra **541** (2020), 126–145, DOI 10.1016/j.jalgebra.2019.09.013. MR4014733

[BT09] Anders Björner and Martin Tancer, *Note: Combinatorial Alexander duality—a short and elementary proof*, Discrete Comput. Geom. **42** (2009), no. 4, 586–593, DOI 10.1007/s00454-008-9102-x. MR2556456

[Coo10] David Cook II, *Simplicial decomposability*, J. Softw. Algebra Geom. **2** (2010), 20–23, DOI 10.2140/jsag.2010.2.20. MR2881131

BEN HERSEY: DEPARTMENT OF MATHEMATICS AND STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ONTARIO, K7L 3N6; `b.hersey@queensu.ca`

GREGORY G. SMITH: DEPARTMENT OF MATHEMATICS AND STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ONTARIO, K7L 3N6, CANADA; `ggsmith@mast.queensu.ca`

ALEXANDRE ZOTINE: DEPARTMENT OF MATHEMATICS AND STATISTICS, QUEEN'S UNIVERSITY, KINGSTON, ONTARIO, K7L 3N6; `18az45@queensu.ca`